NAG Fortran Library Routine Document

F01RKF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F01RKF returns the first ℓ rows of the n by n unitary matrix P^H, where P is given as the product of Householder transformation matrices.

This routine is intended for use following F01RJF.

2 Specification

```
SUBROUTINE F01RKF(WHERET, M, N, NROWP, A, LDA, THETA, WORK, IFAIL)
INTEGER            M, N, NROWP, LDA, IFAIL
complex            A(LDA,*), THETA(*), WORK(*)
CHARACTER*1        WHERET
```

3 Description

P is assumed to be given by

$$P = P_m P_{m-1} \cdots P_1,$$

where

$$P_k = I - \gamma_k u_k u_k^H,$$

$$u_k = \begin{pmatrix} w_k \\ \zeta_k \\ 0 \\ z_k \end{pmatrix},$$

γ_k is a scalar for which $\mathrm{Re}\,\gamma_k = 1.0$, ζ_k is a real scalar, w_k is a $(k-1)$ element vector and z_k is an $(n-m)$ element vector. w_k must be supplied in the kth row of A in elements $\mathrm{A}(k,1),\ldots,\mathrm{A}(k,k-1)$. z_k must be supplied in the kth row of A in elements $\mathrm{A}(k,m+1),\ldots,\mathrm{A}(k,n)$ and θ_k, given by

$$\theta_k = (\zeta_k, \mathrm{Im}\,\gamma_k),$$

must be supplied either in $\mathrm{A}(k,k)$ or in $\mathrm{THETA}(k)$, depending upon the parameter WHERET.

4 References

Golub G H and van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

Wilkinson J H (1965) *The Algebraic Eigenvalue Problem* Oxford University Press, Oxford

5 Parameters

1: WHERET – CHARACTER*1 *Input*

On entry: indicates where the elements of θ are to be found as follows.

If WHERET = 'I' (In A), the elements of θ are in A.

If WHERET = 'S' (Separate), the elements of θ are separate from A, in THETA.

Constraint: WHERET must be one of 'I' or 'S'.

2: M – INTEGER *Input*

On entry: m, the number of rows of A.

Constraint: M $\geq$ 0.

3: N – INTEGER *Input*

On entry: n, the number of columns of A.

Constraint: N $\geq$ M.

4: NROWP – INTEGER *Input*

On entry: the required number of rows of P, ℓ.

When NROWP=0 then an immediate return is effected.

Constraint: $0 \leq$ NROWP $\leq$ N.

5: A(LDA,*) – ***complex*** array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, \mathrm{N})$.

On entry: the leading m by m strictly lower triangular part of the array A, and the m by $(n-m)$ rectangular part of A with top left-hand corner at element $\mathrm{A}(1, \mathrm{M}+1)$ must contain details of the matrix P. In addition, when WHERET = 'I', then the diagonal elements of A must contain the elements of θ.

On exit: the first NROWP rows of the array A are overwritten by the first NROWP rows of the n by n unitary matrix P^H.

6: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F01RKF is called.

Constraint: LDA $\geq \max(1, \mathrm{M}, \mathrm{NROWP})$.

7: THETA($*$) – ***complex*** array *Input*

Note: the dimension of the array THETA must be at least $\max(1, \mathrm{M})$ when WHERET = 'S'.

On entry: with WHERET='S', the array THETA must contain the elements of θ. If THETA(k) = 0.0 then P_k is assumed to be I, if THETA(k) = α, with $\mathrm{Re}(\alpha) < 0.0$ then P_k is assumed to be of the form

$$P_k = \begin{pmatrix} I & 0 & 0 \\ 0 & \alpha & 0 \\ 0 & 0 & I \end{pmatrix},$$

otherwise THETA(k) is assumed to contain θ_k given by

$$\theta_k = (\zeta_k, \mathrm{Im}\, \gamma_k).$$

When WHERET = 'I' or 'i', the array THETA is not referenced.

8: WORK($*$) – ***complex*** array *Workspace*

Note: the dimension of the array WORK must be at least $\max(\mathrm{M}-1, \mathrm{NROWP}-\mathrm{M}, 1)$.

9: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= -1$

On entry, WHERET $\neq$ 'I' or 'S',
or M < 0,
or N $<$ M,
or NROWP < 0 or NROWP $>$ N,
or LD $< \max(1, \mathrm{M}, \mathrm{NROWP})$.

7 Accuracy

The computed matrix P satisfies the relation

$$P = Q + E,$$

where Q is an exactly unitary matrix and

$$\|E\| \leq c\epsilon,$$

ϵ the ***machine precision*** (see X02AJF), c is a modest function of n, and $\|.\|$ denotes the spectral (two) norm. See also Section 7 of the document for F01RJF.

8 Further Comments

The approximate number of floating-point operations is given by

$$\tfrac{8}{3}n[(3n-m)(2\ell-m) - m(\ell-m)], \quad \text{if } \ell \geq m, \text{ and}$$

$$\tfrac{8}{3}\ell^2(3n-\ell), \quad \text{if } \ell < m.$$

9 Example

To obtain the 5 by 5 unitary matrix P following the RQ factorization of the 3 by 5 matrix A given by

$$A = \begin{pmatrix} -0.5i & 0.4-0.3i & 0.4 & 0.3 & 0.4i & 0.3i \\ -0.5-1.5i & 0.9-1.3i & -0.4-0.4i & 0.1-0.7i & 0.3-0.3i \\ -1.0-1.0i & 0.2-1.4i & 1.8 & 0.0 & -2.4i \end{pmatrix}.$$

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F01RKF Example Program Text
*     Mark 14 Release.  NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          MMAX, NMAX, LDA, LDPH
      PARAMETER        (MMAX=10,NMAX=20,LDA=MMAX,LDPH=NMAX)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, J, M, N, NROWP
```

```
*     .. Local Arrays ..
      complex           A(LDA,NMAX), PH(LDPH,NMAX), THETA(NMAX),
     +                  WORK(NMAX)
*     .. External Subroutines ..
      EXTERNAL          F01RJF, F01RKF
*     .. Intrinsic Functions ..
      INTRINSIC         conjg
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F01RKF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) M, N
      WRITE (NOUT,*)
      IF ((M.GT.MMAX) .OR. (N.GT.NMAX)) THEN
         WRITE (NOUT,*) 'M or N is out of range.'
         WRITE (NOUT,99999) 'M = ', M, '   N = ', N
      ELSE
         READ (NIN,*) ((A(I,J),J=1,N),I=1,M)
         IFAIL = 0
*
*        Find the RQ factorization of A
         CALL F01RJF(M,N,A,LDA,THETA,IFAIL)
*
*        Copy the array A into PH and form the n by n matrix conjg(P')
         DO 40 J = 1, N
            DO 20 I = 1, M
               PH(I,J) = A(I,J)
   20       CONTINUE
   40    CONTINUE
         NROWP = N
         IFAIL = 0
*
         CALL F01RKF('Separate',M,N,NROWP,PH,LDPH,THETA,WORK,IFAIL)
*
         WRITE (NOUT,*) 'Matrix  P'
         DO 60 I = 1, N
            WRITE (NOUT,99998) (conjg(PH(J,I)),J=1,NROWP)
   60    CONTINUE
      END IF
      STOP
*
99999 FORMAT (1X,A,I5,A,I5)
99998 FORMAT (5(' (',F6.3,',',F6.3,')',:))
      END
```

9.2 Program Data

```
F01RKF Example Program Data
  3    5                                   :Values of M and N
( 0.00,-0.50) ( 0.40,-0.30) ( 0.40, 0.00) ( 0.30, 0.40) ( 0.00, 0.30)
(-0.50,-1.50) ( 0.90,-1.30) (-0.40,-0.40) ( 0.10,-0.70) ( 0.30,-0.30)
(-1.00,-1.00) ( 0.20,-1.40) ( 1.80, 0.00) ( 0.00, 0.00) ( 0.00,-2.40)
                                           :End of matrix A
```

9.3 Program Results

```
 F01RKF Example Program Results

 Matrix  P
 (-0.197, 0.197) ( 0.164,-0.492) ( 0.277,-0.277) ( 0.364, 0.321) ( 0.012, 0.514)
 ( 0.039, 0.276) (-0.295,-0.426) (-0.055,-0.388) (-0.475, 0.098) (-0.419,-0.299)
 ( 0.315,-0.158) ( 0.452,-0.320) (-0.499,-0.000) (-0.276,-0.305) (-0.034, 0.387)
 ( 0.197,-0.591) (-0.047,-0.331) ( 0.000, 0.000) ( 0.512,-0.047) (-0.361,-0.324)
 (-0.118,-0.565) ( 0.033, 0.208) (-0.000,-0.666) (-0.229, 0.207) ( 0.290, 0.025)
```